

Functional Programming

WS 2012/13

Harald Zankl (VO)


Cezary Kaliszyk (PS) Thomas Sternagel (PS)

Computational Logic
Institute of Computer Science
University of Innsbruck

week 2



User-defined Types



Recursion, Pattern Matching, and Currying

Example

```
let rec map (f,ls) = match ls with
| Nil          -> Nil
| Cons(x,xs)  -> Cons(f x,map(f,xs))
```

Recursion, Pattern Matching, and Currying (cont'd)

Currying

```
let rec map f ls = match ls with
| Nil          -> Nil
| Cons(x,xs)  -> Cons(f x,map f xs)
```

Syntactic Sugar

```
let rec map f = function
| []          -> []
| x::xs      -> f x::map f xs
```

This Week

Practice I

OCaml introduction, **lists**, strings, trees

Theory I

lambda-calculus, evaluation strategies, induction, reasoning about functional programs

Practice II

efficiency, tail-recursion, combinator-parsing

Theory II

type checking, type inference

Advanced Topics

lazy evaluation, infinite data structures, monads, . . .

The Type of Lists

Polymorphic Lists

```

type 'a list = [] | (::) of ('a * 'a list)
           predefined type      infix 'cons'

```

Example

<code>[true;false]</code>	<code>bool list</code>	<code>[(3,2)]</code>	<code>(int * int)list</code>
<code>[1;3;5;7]</code>	<code>int list</code>	<code>[]</code>	<code>'a list</code>
<code>['a';'b']</code>	<code>char list</code>	<code>2::3::[]</code>	<code>int list</code>
<code>["str";"w"]</code>	<code>string list</code>	<code>'a'::['b']</code>	<code>char list</code>

Accessing List Elements - Selectors

```
let hd = function x::_ -> x
          | _      -> failwith "empty_list"

let tl = function _::xs -> xs
          | _          -> failwith "empty_list"
```

A Polymorphic List Function

Example (Replicate)

```
let rec replicate n x =
  if n < 1 then [] else x::replicate (n-1) x
```

- ▶ this function has type `int -> 'a -> 'a list`
- ▶ hence it is polymorphic in `x`

```
replicate 2 'c'
→ if 2 < 1 then [] else 'c'::replicate (2-1) 'c'
→+ 'c'::replicate 1 'c'
→ 'c'::if 1 < 1 then [] else 'c'::replicate (1-1) 'c'
→+ 'c'::'c'::replicate 0 'c'
→ 'c'::'c'::if 0 < 1 then []
                               else 'c'::replicate (0-1) 'c'
→+ ['c';'c']
```


Functions on Integer Lists

Example (Range, Sum, Prod)

```
let rec range m n = if m >= n then []  
                    else m::range (m+1) n
```

```
let rec sum = function []    -> 0  
                    | x::xs -> x + sum xs
```

```
let rec prod = function []    -> 1  
                    | x::xs -> x * prod xs
```

Functions on Integer Lists (cont'd)

```
range 1 3 = [1;2]
```

```
range 3 2 = []
```

```
sum [1;2;3] = 1 + 2 + 3
```

```
sum [] = 0
```

```
prod [1;2;3] = 1 * 2 * 3
```

```
prod [] = 1
```

$$\text{sum}(\text{range } 1 \ n) = \sum_{i=1}^{n-1} i$$

Higher-Order Functions

- ▶ functions taking functions as arguments

Example (Map)

```
let rec map f = function [] -> []  
                      | x::xs -> f x::map f xs
```

```
map succ [1;2;3]  
→ succ 1::map succ [2;3]  
→ 2::map succ [2;3]  
→ 2::succ 2::map succ [3]  
→ 2::3::map succ [3]  
→ 2::3::succ 3::map succ []  
→ 2::3::4::map succ []  
→ 2::3::4::[] = [2;3;4]
```

Fold - A Very Expressive Function

```
let rec foldr f b = function [] -> b
                          | x::xs -> f x (foldr f b xs)

foldr f b [e1;e2;...;eN] = f e1 (f e2 (...(f eN b)...))
```

```
let sum xs = Lst.foldr ( + ) 0 xs
```

```
let prod xs = Lst.foldr ( * ) 1 xs
```

Structuring Code

Modules are used to ...

- ▶ split source code into several files
- ▶ separate namespaces for functions and types
- ▶ abstract from concrete representations

Module Basics - Split Source Code

- ▶ for each module *Module* create **implementation** file *module.ml*
- ▶ code of each module goes into corresponding *.ml file

Example

```
let hd = ...
let tl = ...
let rec map f = ...
let rec foldr f b = ...
```

Lst.ml

```
let rec range m n = ...
let sum = ...
let prod = ...
```

IntLst.ml

Module Basics - Separate Namespaces

- ▶ refer to function *fun* from module *Module* by *Module.fun*
- ▶ no problem to have same function names in different modules

Example

Compute greatest number that can be encoded in binary using n bits

```
let pow2 n = IntLst.prod(Lst.replicate n 2)
```

`Int.ml`

```
let main() =  
  let n = read_int() in  
  let r =  
    IntLst.sum (Lst.map Int.pow2 (IntLst.range 0 n)) in  
  Printf.printf "%i\n" r  
in main()
```

`maxbin.ml`

Module Basics - Abstraction

- ▶ create **interface** file `module.mli` for module *Module*
- ▶ **signature** (i.e., names and types) of module goes into corresponding `*.mli` file

Example

```

type 'a t
val empty : 'a t
val push : 'a -> 'a t -> 'a t
val pop : 'a t -> ('a * 'a t)

```

stck.mli

```

open Lst
type 'a t = 'a list
let empty = []
let push e s = e::s
let pop s = (hd s,tl s)

```

stck.ml

```

type 'a t =
  | Empty
  | Full of ('a * 'a t)
...

```

stck.ml